' Update Class Info 1- © 1993-2002 Aiman Hanna

' This program explores more of file operations. In specific the program introduces a different 

' way of reading and writing files. 

' In addition, the program shows how messages boxes can be used to confirm an 

' operation (Yes/No answer). 

' Key Points:

'       1) Open files for input (read) and output (write)

'       2) EOF() function 

'       3) Getting answers from message boxes

Option Base 1

Option Explicit

Const maxstudnum As Integer = 30

Dim rowsfilled As Integer

Dim clarr(maxstudnum, 4) As String    ' Array of 30 rows and 4 columns

Private Sub form_load()

    ' Once the form is loaded, read the input file into

    ' the clarr array

    Dim str As String, fname As String, gname As String

    Dim id As String, marks As Single

    Dim filename As String

    Dim tempstr1 As String, tempstr2 As String

    Dim spacepos As Integer

    Dim i As Integer, j As Integer

    filename = "C:\comp201\c201_info.dat"

    ' Open the file for input (to read from)

    Open filename For Input As #1

    ' Read from the file until you reach its end. Insert all information

    ' into the array

    Do Until EOF(1)

        Input #1, str

        i = i + 1

        ' Extract the 4 fields that compose every line

         spacepos = InStr(str, " ")

         tempstr1 = Mid$(str, spacepos)

         id = Left$(str, Len(str) - Len(tempstr1))

         tempstr1 = Trim$(tempstr1)

         spacepos = InStr(tempstr1, " ")

         tempstr2 = Mid$(tempstr1, spacepos)

         fname = Left$(tempstr1, Len(tempstr1) - Len(tempstr2))

         tempstr2 = Trim$(tempstr2)

         ' At this point, tempstr2 has the given name and the marks

         spacepos = InStr(tempstr2, " ")

         tempstr1 = Mid$(tempstr2, spacepos)

         gname = Left$(tempstr2, Len(tempstr2) - Len(tempstr1))

         ' marks can simply be found by triming tempstr1 at this point

         marks = Trim$(tempstr1)

         ' Now all fields are obtained for one line. Enter them into the array

         clarr(i, 1) = id

         clarr(i, 2) = fname

         clarr(i, 3) = gname

         clarr(i, 4) = marks

    Loop

   ' Notice that the value of i indicates how many rows of the array are actually

   ' filled. We must keep track of this value

   rowsfilled = i

    ' At this point the file is totally read, close it

    Close #1

End Sub

Private Sub cmdRecordEntry_Click()

    Dim fname As String, gname As String

    Dim id As String, marks As Single

    ' Read the entered information and add an entry to the array.

    ' Notice that the program should perform a mass of validations here, however,

    ' to keep this program as simple as possible, it is assumed that the user will

    ' enter all needed, and correct, information. Based on this assumption, it is

    ' normal that the program misbehaves if the user misbehaves!

    fname = txtFN.Text

    gname = txtGN.Text

    id = cmbID.Text

    marks = txtMarks.Text

    ' Add the entry at the first empty row in the array

    rowsfilled = rowsfilled + 1

    clarr(rowsfilled, 1) = id    

    clarr(rowsfilled, 2) = fname
' You could simply write: clarr(rowsfilled, 2) = txtFN.Text
    clarr(rowsfilled, 3) = gname

    clarr(rowsfilled, 4) = marks

    ' Notice that we only added the entry to the array. Nothing is done to save this

    ' entry to the file yet. This will occur when you click on the Save button.

End Sub

Private Sub cmdSave_Click()

    ' To save the updates, we open the file for writing and then copy the array to

    ' the file. The function will confirm first before updating the file.

    Dim ans As Integer

    Dim i As Integer, j As Integer

    Dim filename As String

    Dim str As String

    ans = MsgBox("Are you sure you want to save your updates?", vbYesNo, "Confirm Saving Request")

    If ans = vbNo Then

        Exit Sub

    Else

        ' open the file for writing

        filename = "C:\comp201\c201_info.dat"

        Open filename For Output As #1

        str = ""

        For i = LBound(clarr) To rowsfilled

            For j = LBound(clarr, 2) To UBound(clarr, 2)

                str = str & clarr(i, j) & "  "

            Next j

            ' Finally place one line into the file

            Write #1, str

            str = ""

        Next i

        ' File has been updated. Close it

        Close #1

    End If

End Sub

Private Sub cmdShowClassInfo_Click()

    Dim i As Integer, j As Integer

    lstClassInfo.Clear

    cmbID.Clear

    For i = LBound(clarr) To rowsfilled

            lstClassInfo.AddItem clarr(i, 1) & "  " & clarr(i, 2) _

                                 & "  " & clarr(i, 3) & "  " & clarr(i, 4)

            ' If you want to align what will be displayed, comment out the above 2 line

            ' and uncomment the following 3 line

            ' lstClassInfo.AddItem clarr(i, 1) & Format$(clarr(i, 2), String$(15, "@")) _

            '                     & Format$(clarr(i, 3), String$(15, "@")) _

            '                     & Format$(clarr(i, 4), String$(15, "@"))

            ' insert the ids into the combo box

            cmbID.AddItem clarr(i, 1)

    Next i

End Sub

Private Sub cmdExit_Click()

    End

End Sub

